AD A068084

(12)
NW

RADC-TR-79-3
In-House Report
January 1979

LEVEL II

HILBERT TRANSFORM BY NUMERICAL INTEGRATION

I. J. Weinberg

DDC FILE COPY

APPROVED FOR PUBLIC RELEASE; DISTRIBUTION UNLIMITED

ROME AIR DEVELOPMENT CENTER
Air Force Systems Command
Griffiss Air Force Base, New York 13441

This report has been reviewed by the RADC Information Office (OI) and is releasable to the National Technical Information Service (NTIS). At NTIS it will be releasable to the general public, including foreign nations.

RADC-TR-79-3 has been reviewed and is approved for publication.

APPROVED: Walter Rotman

WALTER ROTMAN
Chief, Antennas & RF Components Branch

APPROVED: Allan C. Schell

ALLAN C. SCHELL
Chief, Electromagnetic Sciences Division

FOR THE COMMANDER: John P. Huss

JOHN P. HUSS
Acting Chief, Plans Office

If your address has changed or if you wish to be removed from the RADC mailing list, or if the addressee is no longer employed by your organization, please notify RADC (EEA), Hanscom AFB MA 01731. This will assist us in maintaining a current mailing list.

Do not return this copy. Retain or destroy.

MISSION
of
Rome Air Development Center

RADC plans and conducts research, exploratory and advanced development programs in command, control, and communications (C^3) activities, and in the C^3 areas of information sciences and intelligence. The principal technical mission areas are communications, electromagnetic guidance and control, surveillance of ground and aerospace objects, intelligence data collection and handling, information system technology, ionospheric propagation, solid state sciences, microwave physics and electronic reliability, maintainability and compatibility.

Printed by
United States Air Force
Hanscom AFB, Mass. 01731

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER (14) RADC-TR-79-3	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) (6) HILBERT TRANSFORM BY NUMERICAL INTEGRATION		5. TYPE OF REPORT & PERIOD COVERED In-House Report
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) (10) I. J. Weinberg		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Deputy for Electronic Technology (RADC/EEA) Hanscom AFB Massachusetts 01731		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS [redacted] (16) 2305[redacted] (17) J5
11. CONTROLLING OFFICE NAME AND ADDRESS Deputy for Electronic Technology (RADC/EEA) Hanscom AFB Massachusetts 01731		12. REPORT DATE (11) Jan[redacted]79
		13. NUMBER OF PAGES 19
14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office) (12) 19p.		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

*I. J. Weinberg is on a RADC Post Doctoral Program appointment. He is also Professor of Mathematics, University of Lowell, Lowell, MA 01854.

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Hilbert transform
Numerical integration

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

Many engineering studies require the evaluation of a Hilbert Transform of a tabular function. The function is also considered as an even function in order to conform to physical reality. A numerical procedure, incorporated into a Fortran subroutine HTRAN, has been developed to accomplish this task. The procedure has been designed to avoid the theoretical and numerical singularities and employs cubic polynomial interpolation to enhance numerical accuracy.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

309 050 Gm

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

Preface

The author is grateful to James C. Sethares of RADC/EEA for valuable suggestions during the progress of the work.

Contents

Illustrations

Tables

Hilbert Transform by Numerical Integration

1. INTRODUCTION

The Fortran subroutine HTRAN computes the Hilbert Transform of a tabular function of frequency by numerical integration. The determination of the Hilbert Transform is particularly useful in engineering applications such as the computation of the complex impedance when the real or imaginary part is known. In such instances, the real and imaginary parts are related by the Hilbert Transform.

In the physical problems under consideration we can consider the tabular function of frequency as arising from a Fourier Transform of a causal time system, that is, a function of time which is zero for $t < 0$. Such a system gives rise to a complex Fourier Transform with a real part that is an even function of frequency and an imaginary part that is an odd function of frequency. These two parts are then related by the Hilbert Transform. Our tabular function is considered the even function of frequency, its Hilbert Transform is considered the odd function of frequency, and the complex combination of the two are considered the Fourier Transform of a causal system.

(Received for publication 23 January 1979)

2. MATHEMATICAL ANALYSIS

For a function of angular frequency, $R(\omega)$, the Hilbert Transform is defined as

$$X(\omega) = \frac{1}{\pi} \int_{-\infty}^{\infty} \frac{R(\omega')}{\omega' - \omega} \, d\omega' \quad . \tag{1}$$

The integral is to be interpreted as the Cauchy Principal Value so that the singularity arising in the denominator may be eliminated, that is

$$X(\omega) = \lim_{\epsilon \to 0} \left[\int_{-\infty}^{\omega-\epsilon} \frac{R(\omega')}{\omega' - \omega} \, d\omega' + \int_{\omega+\epsilon}^{\infty} \frac{R(\omega')}{\omega' - \omega} \, d\omega' \right] \quad . \tag{2}$$

When the given function is a function of frequency, $R(f)$, we have

$$X(f) = \frac{1}{\pi} \int_{-\infty}^{\infty} \frac{R(f')}{f' - f} \, df' \tag{3}$$

which is interpreted as

$$X(f) = \lim_{\epsilon \to 0} \left[\int_{-\infty}^{f-\epsilon} \frac{R(f')}{f' - f} \, df' + \int_{f+\epsilon}^{\infty} \frac{R(f')}{f' - f} \, df' \right] \quad . \tag{4}$$

We seek to evaluate (3) when $R(f)$ is in tabular form and is an even function of frequency. It is also assumed that $R(f)$ is described tabularly with equal spacing in the frequency axis.

Since we will employ numerical integration, we first modify expression (3) in order to eliminate the singularity. We note, that

$$\int_{-\infty}^{\infty} \frac{df'}{f' - f} = 0 \tag{5}$$

in the sense of the Cauchy Principle Value since

$$\lim_{\epsilon \to 0} \left[\int_{-\infty}^{f-\epsilon} \frac{df'}{f' - f} + \int_{f+\epsilon}^{\infty} \frac{df'}{f' - f} \right] = 0 \quad . \tag{6}$$

Thus any multiple of (5) may be added to (3). In particular, we may write

$$X(f) = \frac{1}{\pi} \int_{-\infty}^{\infty} \frac{R(f') - R(f)}{f' - f} \, df' \tag{7}$$

thus making the singularity in the integrand apparent. The integrand does not become arbitrarily large anywhere and is now suitable for numerical integration.

Since R(f) is an even function we have

$$R(-f) = R(f) \quad . \tag{8}$$

We assume R(f) to be zero outside the range of its tabular description. Denoting f_1 as the first frequency and f_N as the last frequency in the table we have

$$R(f) = 0 \quad . \qquad \left\{ \begin{array}{l} 0 \leq f < f_1 \\ f_N < f \end{array} \right\} \tag{9}$$

We may now write (7) as

$$X(f) = \frac{1}{\pi} \left[\int_{-\infty}^{-f_N} \frac{-R(f)}{f' - f} \, df' + \int_{-f_N}^{-f_1} \frac{R(f') - R(f)}{f' - f} \, df' + \int_{-f_1}^{f_1} \frac{-R(f)}{f' - f} \, df' \right.$$
$$\left. + \int_{f_1}^{f_N} \frac{R(f') - R(f)}{f' - f} \, df' + \int_{f_N}^{\infty} \frac{-R(f)}{f' - f} \, df' \right] \tag{10}$$

where our interest in f values is the range $f_1 \le f \le f_N$. Employing the evenness property, we are able to obtain

$$X(f) = \frac{2f}{\pi}\left[\int_0^{f_1} \frac{-R(f)}{f'^2 - f^2}\,df' + \int_{f_1}^{f_N} \frac{R(f') - R(f)}{f'^2 - f^2}\,df' + \int_{f_N}^{\infty} \frac{-R(f)}{f'^2 - f^2}\,df'\right] . \tag{11}$$

The first and last integrals can be accomplished directly. Thus

$$X(f) = \frac{R(f)}{\pi}\ln\left[\frac{\left(1 - \frac{f}{f_N}\right)(f + f_1)}{\left(1 + \frac{f}{f_N}\right)(f - f_1)}\right] + \frac{2f}{\pi}\int_{f_1}^{f_N} \frac{R(f') - R(f)}{f'^2 - f^2}\,df' . \tag{12}$$

We are henceforth interested in the evaluation of the integral in (12), namely

$$Y(f) = \int_{f_1}^{f_N} \frac{R(f') - R(f)}{f'^2 - f^2}\,df' \tag{13}$$

where the f values are in the range $f_1 \le f \le f_N$.

3. NUMERICAL ANALYSIS

Although the integrand in (13) does not become arbitrarily large anywhere, we are faced with an indeterminacy when f' equals f. The integration technique presently described has the feature that f' never attains any of the f values at which Y is being calculated, thus avoiding the indeterminacy in the integrand.

Consider R(f) defined tabularly as pictured by the solid lines in Figure 1. (f_i, R_i) $i = 1, 2, \ldots, N$ are given with the f values equally spaced. The subroutine computes the Hilbert Transform at these same f values.

Now consider another set of f values, denoted by $\overline{f}_i$, $i = 1, 2, \ldots, N-1$, each of which is midway between two f values. Thus

$$\overline{f}_i = \frac{f_i + f_{i+1}}{2} \qquad i = 1, 2, \ldots, N-1 \tag{14}$$

as pictured by the broken lines in Figure 1. We first apply (12) to obtain the Hilbert Transform, X, at these $\overline{f}$ values where, in the numerical integration, f'



Figure 1. Values of the Independent and Dependent Variables Employed in the Numerical Integration Scheme

ranges over the f values. Thus the indeterminancy will be avoided since f' will never equal one of the $\overline{f}$ values.

To maintain sufficient accuracy in the numerical results, we employ accurate cubic interpolation formulas as part of the numerical scheme.

First, we need to obtain the values $R(\overline{f}_i)$ $i = 1, 2, \ldots, N-1$ which we denote as $\overline{R}_i$ $i = 1, 2, \ldots, N-1$. Employing cubic interpolation for the interior values and parabolic interpolation for the end values we obtain

$$\overline{R}_1 = (3R_1 + 6R_2 - R_3)/8$$

$$\overline{R}_i = (-R_{i-1} + 9R_i + 9R_{i+1} - R_{i+2})/16 \qquad i = 2, 3, \ldots, N-2$$

$$\overline{R}_{N-1} = (-R_{N-2} + 6R_{N-1} + 3R_N)/8 \quad . \tag{15}$$

We now write (13) as

$$\bar{Y}_i = Y(\bar{f}_i) = \int_{f_1}^{f_N} \frac{R(f') - R(\bar{f}_i)}{f'^2 - \bar{f}_i^2}\, df' \qquad i = 1, 2, \ldots, N-1 \tag{16}$$

which we write as, employing numerical integration,

$$\bar{Y}_i = \sum_{j=1}^{N} \frac{h_j(R_j - \bar{R}_i)}{f_j^2 - \bar{f}_i^2} \qquad i = 1, 2, \ldots, N-1 \tag{17}$$

where, when N is odd, Simpson's Rule gives

$$\begin{aligned} h_1 &= h_N = \Delta f/3 \\ h_j &= 4\Delta f/3 \qquad j = 2, 4, \ldots, N-1 \quad (j \text{ even}) \\ h_j &= 2\Delta f/3 \qquad j = 3, 5, \ldots, N-2 \quad (j \text{ odd}) \end{aligned} \tag{18}$$

and Δf is the spacing in the frequency axis.

When N is even the Trapezoidal Rule is used to include the last interval. Thus

$$\begin{aligned} h_1 &= \Delta f/3 \\ h_j &= 4\Delta f/3 \qquad j = 2, 4, \ldots, N-2 \quad (j \text{ even}) \\ h_j &= 2\Delta f/3 \qquad j = 3, 5, \ldots, N-3 \quad (j \text{ odd}) \\ h_{N-1} &= 5\Delta f/6 \\ h_N &= \Delta f/2 \quad . \end{aligned} \tag{19}$$

The Hilbert Transform at the $\bar{f}_i$ $i = 1, 2, \ldots, N-1$ are obtained from (12) as

$$\bar{X}_i = X(\bar{f}_i) = \frac{\bar{R}_i}{\pi} \ln \left[\frac{\left(1 - \frac{\bar{f}_i}{f_N}\right)(\bar{f}_i + f_1)}{\left(1 + \frac{\bar{f}_i}{f_N}\right)(\bar{f}_i - f_1)} \right] + \frac{2\bar{f}_i \bar{Y}_i}{\pi} \quad . \qquad i = 1, 2, \ldots, N-1 \tag{20}$$

To obtain the Hilbert Transform, X_i, at the f_i $i = 1, 2, \ldots, N$ we again perform an accurate interpolation; cubic interpolation for the interior values and parabolic interpolation for two values at each end. Thus

$$
\begin{aligned}
X_1 &= (15\overline{X}_1 - 10\overline{X}_2 + 3\overline{X}_3)/8 \\
X_2 &= (3\overline{X}_1 + 6\overline{X}_2 - \overline{X}_3)/8 \\
X_i &= (-\overline{X}_{i-2} + 9\overline{X}_{i-1} + 9\overline{X}_i - \overline{X}_{i+1})/16 \qquad i = 3, 4, \ldots, N-2 \\
X_{N-1} &= (-\overline{X}_{N-3} + 6\overline{X}_{N-2} + 3\overline{X}_{N-1})/8 \\
X_N &= (3\overline{X}_{N-3} - 10\overline{X}_{N-2} + 15\overline{X}_{N-1})/8 \quad .
\end{aligned}
\tag{21}
$$

We thus obtain the tabular results (f_i, X_i) $i = 1, 2, \ldots, N$ as the Hilbert Transform values for the given (f_i, R_i) $i = 1, 2, \ldots, N$. The only assumption made concerning the function $R(f)$ is that it is an even function of frequency, complying with physical reality. It is also assumed that the function $R(f)$ is described tabularly with equal spacing in f. To insure sufficient accuracy in the numerical integrations and interpolations, one should make the frequency interval adequately small or, correspondingly, N sufficiently large.

4. SUBROUTINE DESCRIPTION

The user employs the Hilbert Transforms subroutine HTRAN by the statement

```
CALL HTRAN (R, X, N, FBEG, FEND)
```

The arguments in the subroutine are described as follows:

- R - a dimensioned array containing the tabular values of R.
- X - a dimensioned array containing the Hilbert Transform values of X returned by the subroutine HTRAN.
- N - the number of values contained in the table of R (and X).
- FBEG - the first frequency, f_1, for the R array.
- FEND - the last frequency, f_N, for the R array.

Since equal spacing in the frequency axis is assumed, the items N, FBEG and FEND permit the subroutine to determine the values f_i $i = 1, 2, \ldots, N$.

5. PROGRAM LISTING

The following is the Fortran listing of the subroutine HTRAN as written for the CDC 6600 at Hanscom Field, Massachusetts.

```
      SUBROUTINE HTRAN(R, X, N, FBEG, FEND)
      DIMENSION R(3), X(3)
      PI=3.14159265359
      FDEL=(FEND-FBEG)/(N-1)
      F=FBEG+.5*FDEL
      INC=MOD(N, 2)
      NI=N+INC-1
       NM1=N-1
      NIM2=NI-2
      DO 33 I=1, NM1
      X(I)=0.
      IF (I .EQ. 1) RX=(3.*R(1)+6.*R(2)-R(3))/8.
      IF (I .EQ. NM1) RX=(-R(N-2)+6.*R(NM1)+3.*R (N))/8.
      IF (I .EQ. 1 .OR. I .EQ. NM1) GO TO 20
      RX=(-R(I-1)+9.*R(I)+9.*R(I+1)-R(I+2))/16.
   20 CONTINUE
      FI=FBEG
      DO 28 IP=1, NIM2, 2
      X(I)=X(I)+4.*(R(IP+1)-RX)/(( FI+FDEL)**2-F**2)
     X        +2.*(R(IP  )-RX)/(  FI       **2-F**2)
      FI=FI+2.*FDEL
   28 CONTINUE
      FEN=FEND
      IF(INC .EQ. 0) FEN=FEND-FDEL
      X(I)=X(I)+(R(NI)-RX)/(FEN**2-F**2)
     X        -(R(1 )-RX)/(FBEG**2-F**2)
      X(I)=FDEL/3.*X(I)
      IF(INC .EQ. 1) GO TO 30
      X(I)=X(I)+.5*FDEL*((R(NI)-RX)/(FEN**2-F**2)
     X        +(R(N)-RX)/(FEND**2-F**2))
   30 X(I)=2./PI*F*X(I)+RX/PI*ALOG
     X ((1.-F/FEND)/(1.+F/FEND)*(F+FBEG)/(F-FBEG))
      F=F+FDEL
   33 CONTINUE
      NM2=N-2
      X1=(15.*X(1)-10.*X(2)+3.*X(3)))/8.
       X2=(3.*X(1)+6.*X(2)-X(3))/8.
      DO  31 I=3, NM2
      XT=(-X(I-2)+9.*X(I-1)+9.*X(I)-X(I+1))/16.
      X(I-2)=X1
      X1=X2
      X2=XT
   31 CONTINUE
      X(N)=(15.*X(NM1)-10.*X(NM2)+3.*X(N-3))/8.
      X(N-1)=(3.*X(NM1)+6.*X(NM2)-X(N-3))/8.
      X(N-2)=X2
      X(N-3)=X1
      RETURN
      END
```

6. NUMERICAL EXAMPLES

6.1 Example 1

For the function

$$R(f) = \begin{Bmatrix} 1 & 0 \leq f \leq 1 \\ 0 & 1 < f \end{Bmatrix} \tag{22}$$

with R(f) an even function, the exact expression for the Hilbert Transform in the range $0 \leq f \leq 1$ is known to be

$$X(f) = \frac{1}{\pi} \ell n \left| \frac{1 - f}{1 + f} \right| . \qquad 0 \leq f \leq 1 \tag{23}$$

This example reduces to the trivial case since the integral in (12) vanishes and the first term in (12) is identical with (23).

However, exact agreement with (23) may not be attained by the subroutine since interpolations and extrapolations are employed to find the Hilbert Transform at the specified f values. Using an N value of 501, exact agreement was obtained in the range $0 \leq f \leq 1$ except for values of f near $f = 1$.

6.2 Example 2

For the function

$$R(f) = \begin{Bmatrix} (1 - f^2)^{1/2} & 0 \leq f \leq 1 \\ 0 & 1 < f \end{Bmatrix} \tag{24}$$

with R(f) an even function, the exact expression for the Hilbert Transform in the range $0 \leq f \leq 1$ is known to be

$$X(f) = -f . \qquad 0 \leq f \leq 1 \tag{25}$$

Two cases were examined to illustrate the accuracy of the numerical integrations. Table 1 depicts the comparison of the two cases with the exact results from (25).

We first note an overall increase in accuracy for the case with the smaller spacing along the f axis. This behavior is to be expected with a numerical integration scheme such as the one employed.

Table 1. Comparison of Example 2 With Exact Results for Two Different Spacings in the f Direction

f	exact	Δf = .005	Δf = .002
0	0	$-.9395268 \times 10^{-10}$	$-.2351980 \times 10^{-11}$
0.1	-.1	-.1000026	-.1000007
0.2	-.2	-.2000054	-.2000014
0.3	-.3	-.3000085	-.3000022
0.4	-.4	-.4000123	-.4000031
0.5	-.5	-.5000172	-.5000044
0.6	-.6	-.6000242	-.6000061
0.7	-.7	-.7000354	-.7000090
0.8	-.8	-.8000572	-.8000145
0.9	-.9	-.9001214	-.9000309
1.0	-1.0	-1.014396	-1.009106

We also note a decrease in accuracy in both cases as f increases from 0 to 1. This may be explained by the nature of the R(f) function (24). The magnitude of the slope of R(f), and consequently also for the integrand in (12), increases greatly as f goes from 0 to 1. This could well affect the accuracy of the numerical integrations.

6.3 Example 3

For the function

$$R(f) = \frac{\sin 2\pi f}{2\pi f} \qquad 0 \le f \tag{26}$$

with R(f) an even function, the exact expression for the Hilbert Transform valid in the range $0 \le f$ is known to be

$$X(f) = \frac{\cos 2\pi f - 1}{2\pi f} . \qquad 0 \le f \tag{27}$$

Two cases were examined here to illustrate the effect of the numerical approximation for ∞. In one case the upper limit of f, f_N, was chosen as 10 with an N of 641 making a spacing $\Delta f = 1/64$. In the second case f_N was chosen as 20 with an N of 1281 thus keeping the same spacing $\Delta f = 1/64$. Table 2 shows the comparison of the two cases with exact results (27) for the upper limit of f being ∞.

Table 2. Comparison of Example 3 With Exact Results for Two Different Approximations for $f_N = \infty$

f	exact	$f_N = 10$	$f_N = 20$
0	0	$.7444819 \times 10^{-4}$	$.7444819 \times 10^{-4}$
0.25	-.6366198	-.6366234	-.6366199
0.50	-.6366198	-.6366276	-.6366206
0.75	-.2122066	-.2122190	-.2122084
1.0	0	$-.1647500 \times 10^{-4}$	$-.2255452 \times 10^{-5}$
1.25	-.1273240	-.1273442	-.1273264
1.50	-.2122066	-.2122311	-.2122094
1.75	-.09094568	-.09097477	-.09094930
2.00	0	$-.3364389 \times 10^{-4}$	$-.4225282 \times 10^{-5}$

As was to be expected, the case with the larger value for f_N produces more accurate results.

7. SUMMARY

The subroutine HTRAN obtains accurate values of the Hilbert Transform of a tabular function of frequency, which is equally spaced in the frequency axis, is an even function of frequency and is zero outside its range of tabular definition. Numerical integrations based on Simpsons Rule in which singularities and indeterminacies are eliminated and cubic polynomial interpolations are employed. Values of the Hilbert Transform are obtained for the same frequency values as are specified in the tabular definition of the function. As is demonstrated in the examples, the frequency spacing needs to be small in order that the numerical integrations and interpolations produce accurate results.

References

1. Bateman Manuscript Project (1954) Table of Integral Transforms, McGraw-Hill.
2. Papoulis, A. (1962) The Fourier Integral and its Applications, McGraw-Hill.
3. Ganguly, A. K., Webb, D. C., and Banks, C. (1978) Complex Radiation Impedance of Microstrip-Excited Magnetostatic-Surface Waves, IEEE Transactions on Microwave Theory and Techniques, pp. 444-447.
4. Wu, H. J. (1977) Ph.D. Thesis, University of Texas at Arlington, Department of Electrical Engineering.